中央财经大学·中国金融发展研究院

名称：投资学第二次作业

组名：GGM Never Die

成员：孟舒晨、郭好格、高雅萌

日期：10/10/2019

Question.1—4 The Impact of Recommendation

```
option ls=80;
libname worklib 'D:\Investment\data';
proc sort data=worklib.Ipos_1996_2000 out=d_ipo;
   by CUSIP;
run;
proc sort data=worklib.Recommendation out=d_recdats;
   by CUSIP;
run;
data d;
   merge d_ipo d_recdats;
   by CUSIP;
run;
*only keep the earliest recommendation;
proc sort data=d;
   by permno ANNDATS;
run;
proc sort data=d nodupkeys;
   by permno;
run;
```

The part of code is to merge the ipo data and recommendation data by CUSIP variable. Note that before the merge operation we have to sort the CUSIP variable.
The part of code is to delete the late recommendation date, so we drop the duplicate permno after sort then by permno and anndats.

```
*merge daily return and table d by permno;
proc sql;
   create table d1 as
   select distinct
   d.ANNDATS,
   d.IRECCD,
   Assignmentreturns.*
      from worklib.Assignmentreturns, d
      where
      Assignmentreturns.permno=d.permno and
      d.ANNDATS-15<=Assignmentreturns.DATE<=d.ANNDATS+15
   order by
   permno,DATE;
run;
```

We use “proc aql ”command to combine the “anndats” and “IRECCD” variables of “d” set with all variables of “assignmentreturns” set and use two conditions to make sure all the five workdays before and after the “anndats” are in the new table “d1”.

```
*get the num_before;
proc means data=d1 noprint;
    var permno;
    by permno;
    output out=d_stats
    n=num_before;
    label num_before= 'num_before';
    where DATE<ANNDATS;
run;
*get rel_day;
data d2;
    merge d1 d_stats(keep=permno num_before);
    by permno;
    if first.permno then
    rel_day=-num_before;
        else
        rel_day= rel_day +1;
        retain rel_day;
run;
*keep (-5,5) data;
data d3;
    set d2;
    if (-5<=rel_day<=5)&num_before>=5;
run;
```

Then we use “proc means” command to get the number of days before the first recommendation date. The initial relative day is “-num_before” when the first permno shows up then we use “plus one” command until the different permno shows up. Finally we keep the relative days between[-5,5].

```
*merge market return and d3 by date and get ar;
data d4;
    set worklib.Market_returns;
    length date 8;
run;
proc sort data=d4;
    by date;
run;
proc sort data=d3;
    by date;
run;
data d5;
    merge d4 d3;
    by date;
    ar=ret-vwretd;
run;
*calculate acr;
proc sort data=d5;
    by permno date;
run;
data d6;
    set d5;
    by permno ANNDATS;
    if first.permno then
    car=ar;
        else
        car=car+ar;
    retain car;
run;
```

We use merge function to add the market returns then get the abnormal returns. And use plus function to get the accumulative abnormal returns.

```
*define the subsamples;
data d7;
    set d6;
    if IRECCD=1 then
    sample="strong buys";
        else
        if IRECCD=2 then
        sample="buys";
            else
        sample="the others";
run;
```

In order to classify the sample, we create a new variable to describe the IRECCD variable.

```
proc sort data=d7;
    by sample rel_day;
run;
proc means data=d7 noprint;
    var ar;
    output out=result
        mean(ar)=mean_ar
        mean(car)=acar
        t(ar)=t_ar
        n(ar)=n_ar;
    by sample rel_day;
run;
```

We use “proc means” function to get the mean, t-statistic and N parameters of variable “ar” and sort them by “sample”, “rel_day” to get the average of “ar”.

```
symbol1
    color=green interpol=spline width=1 value=square;
symbol2
    color=red interpol=spline width=1 value=triangle;
symbol3
    color=blue interpol=spline width=1 value=circle;
axis1
    label=("Event Time")
    order=-5 to 5 by 1
    width=3;
axis2
    label=("Acummulative Abnormal Return")
    order=-0.08 to 0.06 by 0.02
    width=3;
run;
proc gplot data=result;
    plot acar*rel_day=sample/haxis=axis1 vaxis=axis2;
run;
quit;
```

We use the code above to get the graph which y-axis is acummulative abnormal return and x-axis is relative days. We also define different form of three lines which represents different state of recommendation.



Figure 1 cumulative abnormal returns

This figure plots cumulative market-adjusted returns for “strong buys”, “buys” and “the others” sample surrounding the first announcement date of recommendation. Day0 is the event day. The window[-5,5] is lined up at x-axis and the value of y-axis represents average abnormal return for three different samples. Form the graph above we can conclude that after the event day, the abnormal returns of “the others” sample decrease steeply and bounce back again, which maybe reflect the overreaction of people. Another description of curves is that before the day0 the abnormal returns of “buys” sample is greater than “strong buys” ’s while after the event day, returns of “strong buys” is over than “buys” ’s for a while. It can also be interpreted by “overreaction”.

PANEL:Market-adjusted Returns*(MARs)*--*Buys*

Day	Average MAR(%)	t_statistic	N
-5	0.88%	3.77297	875
-4	0.84%	3.97409	877
-3	0.59%	2.60491	877
-2	1.00%	4.44166	877
-1	0.41%	1.69119	877
0	0.09%	0.32918	877
1	0.01%	0.04144	877
2	-0.27%	-1.19673	877
3	0.09%	0.41601	877
4	-0.27%	-1.22237	877
5	0.07%	0.32028	877

PANEL:Market-adjusted Returns*(MARs)*—*Strong Buys*			
Day	**Average MAR(%)**	**t_statistic**	**N**
-5	0.27%	1.07432	899
-4	0.56%	2.94016	901
-3	0.40%	2.17067	901
-2	0.83%	3.82889	901
-1	1.35%	5.56386	901
0	0.53%	2.17358	901
1	0.30%	1.35729	901
2	-0.30%	-1.58679	901
3	-0.34%	-1.84109	901
4	-0.70%	-3.98535	901
5	-0.06%	-0.34708	901

PANEL:Market-adjusted Returns*(MARs)*—*The others*			
Day	**Average MAR(%)**	**t_statistic**	**N**
-5	-1.21%	-1.14568	51
-4	0.24%	0.30945	51
-3	0.02%	0.01903	51
-2	-0.45%	-0.39299	51
-1	-0.51%	-0.72275	51
0	-1.35%	-0.81941	51
1	-1.98%	-2.17071	51
2	-0.50%	-0.46577	51
3	-0.32%	-0.36984	51
4	2.38%	2.02762	51
5	-0.14%	-0.14986	51

The table provides event-study results for the respective three samples. Day0 marks the recommendation date. From the tables above we can infer that average MAR of “buys” is much better than “strong buys” ’s and turns over after day0. The average MAR of “the others” decreases immediately after day0 and bounces back after day2. The rank of number is : buys>strongbuys>the others, which is logical.

Qestion.5 The Impact of The End of the Quiet Period

This part provides event-study codes and results for the entire sample of 1,851 firms. Day 0 marks the end of the IPO quiet period, which is the 26th calendar day following the IPO (or the first trading day thereafter). We define it as “end_date= offer_date +25”. As mentioned above, we have merged table “ipo_1996_2000” “recommendation”, market_returns” and “AssignmentReturns”, and have calculated the entire firms’ market-adjustment returns of every day during 1996 to 2001, so we don’t talk about these codes here.

```
proc sql;
	create table e as
	select distinct
		a.ar,
		a.date,
		iporec.*
	from
		iporec,a
	where
		iporec.permno=a.permno and
		iporec.offer_date-50<=a.date<=iporec.offer_date+50
	order by
		permno,date;
quit;
```

First, we select the distinct we need from tables and find the days near our event day, In order not to miss the trading days, we expand the range to 50 days. For convenience, we define a new variable to represent the event date.

```
*define end_date;
data e;
	set e;
	if ar=. then delete;
	end_date=offer_date+25;
	retain end_date;
run;
```

Second, we do the same thing as above, we use num_before to generate a variable, rel_day, to show up the distance between the event day and this day. The distinct “a.date” we selected doesn’t contain non-trading days, so when we use num_before to count the number of days before the event day, and the rel_day reduces to 0, it is actually the 26^{th} calendar day following the offer day or the first trading day.

```
*get rel_day;
proc means data=e noprint;
	var permno;
	by permno;
	output out=e_stats(drop=_type_ _freq_)
	n=num_before;
	where date<end_date;
run;
```

```
data f;
	merge e e_stats;
	by permno;
	if first.permno then
		rel_day=-num_before;
	else
		rel_day= rel_day +1;
	retain rel_day;
run;
```

Third, we get the rel_day, and sort the data by rel_day, and use the proc means to get the statistics we need. Finally, we get Table II below.

```
data f;
	set f;
	if -5<=rel_day<=5;
run;
```

```
data f;
	set f;
	by permno;
	if first.permno then
		car=ar;
	else
		car=car+ar;
	retain car;
run;
```

```
proc sort data=f;
by rel_day;

proc means data=f noprint;
	var car ar;
	output out=f_stats(drop=_type_ _freq_)
	n(ar)=ar
	mean(ar)=aar
	t(ar)=t_ar
	n(car)=ncar
	mean(car)=acar
	t(car)=t_car;
	by rel_day;
run;
```

Table II: The End of the Quiet Period : MAR

Day	N	MAR(%)	t-stat
-5	1851	0.4879	3.0654
-4	1851	0.3856	2.78286
-3	1851	0.5719	4.12084
-2	1851	0.9273	6.51631
-1	1851	1.5045	9.74843
0	1851	0.5909	3.50394
1	1851	0.1334	0.81858
2	1851	0.1326	0.86058
3	1851	-0.077	-0.55343
4	1851	-0.1506	-1.10844
5	1851	-0.4879	-3.31445

Table II presents the average market-adjusted returns. The T-stats tests the null hypothesis that the percentage of abnormal returns is the same as zero in the estimation period assuming independence. Day 0 is the first trading day after the 25th calendar day since trading commenced. There is a clear clustering of significant abnormal returns in the days surrounding the end of the quiet period. Table II also shows that significant abnormal returns begin to occur several days before the end of the quiet period. The single largest daily abnormal return, 1.50 percent, occurs on day - 1, and abnormal returns are positive and significant on every day in the (- 5,- 1) window. In contrast, abnormal returns decrease to 0.1 percent on day 1, and continue to decrease in the (1, 5) window.

As with the price drops associated with lockup expirations, the significant abnormal returns in Table II at the end of the quiet period appear to **be inconsistent with market efficiency** because the relevant dates are known ahead of time with complete certainty. The fact that the abnormal returns occur in advance of the event date is similarly puzzling.

Question.6 The impact of initiations of coverage

Firstly, we merge file_Ipos and file_Recommendation by their same variable, CUSIP.

```
*question 6;
*merge ipos and recommendation;
proc sort data=worklib.Ipos_1996_2000 out=d_ipo;
    by CUSIP;
run;
proc sort data=worklib.Recommendation out=d_recdats;
    by CUSIP;
run;
data d8;
    merge d_ipo d_recdats;
    by CUSIP;
run;
```

Secondly, we only keep the earliest recommendation for every firm by using the command “NODUPKEYS”, because we need the initiation of coverage.

```
*only keep the earliest recommendation;
proc sort data=d8;
    by permno ANNDATS;
run;
proc sort data=d8 nodupkeys;
    by permno;
run;
```

Thirdly, we merge that table we get in the previous step and firms’ daily returns and then market returns by permno and date respectively. At the same time, we calculate abnormal returns.

```
proc sort data=worklib.Assignmentreturns;
    by permno;
data d8;
    merge d8 worklib.Assignmentreturns;
    by permno;
    format offer_date YYMMDDN8.;
run;
data d_mret;
    set worklib.Market_returns;
    length date 8;
run;
proc sort data=d8;
    by date;
proc sort data=d_mret;
    by date;
data d8;
    merge d8 d_mret;
    by date;
    ar=ret-vwretd;

    informat offer_date YYMMDD6.;
    end_quiet_period=offer_date+25;
    format end_quiet_period YYMMDDN8.;
    informat end_quiet_period YYMMDD6.;
run;
```

Then, we select the dates falling in 20 calendar days before and after the end_quiet_period, which will include more than 10 trading days before and after the end_quiet_period and satisfy our needs for both question 6 and 7.

```
proc sql;
    create table d8 as
    select distinct
    d8.*
    from d8
    where
    end_quiet_period-20<=date<=end_quiet_period+20
    order by permno,end_quiet_period,date;
run;
```

Then, we define the subsamples, “with initiations” and “without initiations”, based on whether on not there are recommendations in the [-2,+2] window around the end of the quiet period.

```
data d8;
    set d8;
    if end_quiet_period-2<=ANNDATS<=end_quiet_period+2 then
    sample="with initiations";
    else
    sample="without initiations";
run;
```

Then, we get the num_before and rel_day. Notice here, we use the command RETAIN, because need to use the previous data when getting rel_day.

```
proc means data=d8 noprint;
    var permno;
    by permno end_quiet_period;
    output out=d_stats4
    n=num_before;
    where date<end_quiet_period;
run;
data d8;
    merge d8 d_stats4;
    by permno end_quiet_period;
    if first.end_quiet_period then
    rel_day=-num_before;
    else
    rel_day=rel_day+1;
    retain rel_day;
run;
```

For question 6, we just need to calculate the cumulative abnormal returns for [-5,+5] event window and the event is the end of quiet period.

```
data d9;
    set d8;
    if -5<=rel_day<=5;
run;
```

Then, we calculate car by using RETAIN command again, since cumulation implies using data from the previous observation.

```
data d9;
    set d9;
    by permno end_quiet_period date;
    if first.end_quiet_period then
    car=ar;
    else
    car=car+ar;
    retain car;
run;
```

Lastly, we get some statistical results for abnormal returns with dividing the total data into two subsamples, that is, “with initiations” and “without initiations”.

```
proc sort data=d9;
    by sample rel_day;
proc means data=d9 noprint;
    var ar;
    output out=d_stats5
    mean(ar)=mean_ar
    t(ar)=t_ar
    n(ar)=n_ar;
    by sample rel_day;
run;
```

And finally, we get the required tables, which include average market adjusted returns, its t-statistic and the number of observations, shown as below.

sample=with initiations			
rel_day	Average MAR(%)	t-stat	N
-5	0.010618112	2.89742	634
-4	0.007733465	2.73025	634
-3	0.009322037	3.18426	634
-2	0.015436971	5.12807	634
-1	0.027292444	8.35535	634
0	0.005059663	1.41571	634
1	0.004662738	1.35871	634
2	0.002249351	0.70764	634
3	-0.001298998	-0.4566	634
4	-0.002280649	-0.8092	634
5	-0.007837911	-2.6573	634

sample=without initiations			
rel_day	Average MAR(%)	t-stat	N
-5	0.000679655	0.48337	1376
-4	0.001143312	0.7984	1376
-3	0.003739033	2.65296	1376
-2	0.006024426	4.10504	1376
-1	0.008324548	5.25455	1376
0	0.005074257	3.09828	1376
1	-0.000494806	-0.3113	1376
2	9.72112E-05	0.06316	1376
3	-0.00081815	-0.5595	1376
4	1.05313E-05	0.00674	1376
5	-0.003272087	-2.0895	1376

According to these two tables, we can find that there are 634 firms had the initiation of coverage surrounding the end of quiet period within the window [-2,+2], which accounts for 31.54% of our total sample. And about 68.46% of firms didn't have such initiations of coverage. The average abnormal return for the 634 firms that have coverage initiated is 5.3 percent during the [- 2,+2]-day event window, which is highly significant. The average abnormal return for the noninitiations category in the same window is 1 percent. We then know clearly the average abnormal returns are bigger and more significant for the firms with initiations of coverage.

Question.7 Plot The Results

For question 7, we need to plot the average cumulative abnormal returns for those two subsamples over a 21-day (-10,+10) window of the end of quiet period.

Firstly, we select the day-window [-10,+10] through the condition, "-**10**<=rel_day<=**10**".

```
*question 7;
data d10;
   set d8;
   if -10<=rel_day<=10;
run;
```

Secondly, we also calculate car which is similar to question 6 and again use the RETAIN command.

```
data d10;
    set d10;
    by permno end_quiet_period date;
    if first.end_quiet_period then
    car=ar;
    else
    car=car+ar;
    retain car;
run;
```

Finally, we plot the figure, as shown below.

```
proc sort data=d10;
    by sample rel_day;
proc means data=d10 noprint;
    var car;
    output out=d_stats6
    mean(car)=acar;
    by sample rel_day;
run;
    symbol1
    color=green interpol=spline width=1 value=square;
    symbol2
    color=red interpol=spline width=1 value=triangle;
    axis1
    label=('event time')
    order=-10 to 10 by 1
    width=3;
    axis2
    label=('Cumulative Abnormal Return')
    order=-0.04 to 0.14 by 0.01
    width=3;
run;
proc gplot data=d_stats6;
    plot acar*rel_day=sample/haxis=axis1 vaxis=axis2;
run;
quit;
```



Figure 2 cumulative abnormal returns

This figure plots cumulative market-adjusted returns for the initiation and noninitiation samples surrounding the end of the quiet period. Day 0 is the event day. The window [- 10,+10] is plotted. The average abnormal return for the 634 firms that have coverage initiated is 5.3 percent during the (- 2,+2)-day event window.The average abnormal return for the noninitiations category in the same event window is 1 percent. IPO data are from January 1,1996, to December 31, 2001.

We can also find from the figure that, the cumulative market-adjusted increase in value of the initiations group is about 11 percent, while the CAR for the noninitiations group over this period is 2 percent.

This figure also observes the pre-event run-up in prices. From day - 10 through day - 1, the average firm with coverage-initiated experiences a CMAR of almost 10 percent. Over the same period, the noninitiations group has a CMAR of almost exactly zero. This result shows that market participants are evidently able to ascertain which firms will ultimately receive coverage.

Two potential (and nonexclusive) explanations for the market's apparent clairvoyance regarding initiations are (1) analyst behavior is highly predictable and (2) there is information leakage. For example, CNBC commentators will on occasion remark that a particular bank is "expected" to initiate coverage after the close of trading and will even suggest what the rating is going to be.

The zero CMAR experienced for firms that do not have coverage initiated is also of interest. Conceivably, analysts tend to not initiate coverage on firms when they are unwilling to issue a positive recommendation. If this were true, then we might expect firms with no coverage to decline in value. We call this the "no news is bad news" hypothesis.

However, our noninitiations sample includes an unknown number of firms that actually do have coverage initiated, but are not identified as such in our data sources, or who initiate more than three days after the end of the quiet period. As a result, the apparent lack of reaction in this sample may be the product of a mixture of firms that do and do not have coverage initiated. Thus, this is a potential issue of our analysis and the result will be more precise if we can solve it.